


Integer Conversion



SVFIG

Sep. 23, 2022

Bill Ragsdale

The Challenge

Write a number to text conversion covering integers 0 to 100.

Conversion Algorithm

1. Factor for case structures.
2. Process by decades 0, 1, 2 ... to 10.
3. Exception: 90 is ninety, 91 is ninety-one.
4. Special handling for 10 to 19.
5. Report out of range error.

Setup

```
CREATE Output.Text 31 allot \ final text
CREATE Scratch 31 allot \ interim text

: concat ( addr1 addr2 addr3 --- )
  \ concatenate two strings into addr3
  2 pick 2dup c@ 1+ cmove \ first part
  over count >r over 4 pick c@
  1+ + r> cmove \ second part
  >r c@ swap c@ + r> c! ; \ final length
```

Units Conversion

```
: do.units ( n --- string2 )
  \ convert 0..9 into a counted string
  10 mod
  case 0 of c" zero"      endof
        1 of c" one"      endof
        2 of c" two"      endof
        3 of c" three"    endof
        4 of c" four"     endof
        5 of c" five"     endof
        6 of c" six"      endof
        7 of c" seven"    endof
        8 of c" eight"    endof
        9 of c" nine"     endof
  abort" units error"    endcase ;
```

Tens Conversion

```
: add.teen ( n --- addr1 addr2 ) do.units c" teen" ;
: do.tens  ( n --- addr1 addr2 )
  dup
  case 10 of drop c" ten"      c" " endof
        11 of drop c" eleven"  c" " endof
        12 of drop c" twelve"  c" " endof
        13 of drop c" thirteen" c" " endof
        14 of add.teen
        15 of drop c" fifteen"  c" " endof
        16 of add.teen
        17 of add.teen
        18 of add.teen
        19 of add.teen
        abort" teens error"    endcase ;
```

Fixup For Decades

```
: do.decade ( n addr1 --- addr1 addr2 )
  \ build hyphenated "nine-one"
  over 10 mod      \ if mod = zero, shorten
  0= if ( 90 etc ) nip c" "
      else \ 9x append '-' to string
           c" -" scratch concat scratch
           swap do.units
      then ;

\ 90 yields "ninety"
\ 91 yields "ninety-one"
```

Top Down By Decade

```
: all.decades ( n --- string1 string2 )
  \ resolve input by decade
  dup 100 > over 0< or abort" out of range"
  dup 10 /
  case 10 of drop
    c" one hundred" c" "      endof
  9 of c" ninety" do.decade  endof
  8 of c" eighty" do.decade  endof
  7 of c" seventy" do.decade endof
  6 of c" sixty" do.decade  endof
  5 of c" fifty" do.decade  endof
  4 of c" forty" do.decade  endof
  3 of c" thirty" do.decade  endof
  2 of c" twenty" do.decade  endof
  1 of          do.tens      endof
  0 of          do.units c" " endof
  abort" decade error"      endcase ;
```


Top Level Elements

```
\ Convert and display one integer.
: convert.one ( n --- )
    all.decades output.text concat
    output.text count type ;

\ Report writer, integers 0 to 100.
: full cr
    101 0 do cr i 4 .r 2 spaces
    i convert.one loop ;
```

Output Example

0	zero	10	ten
1	one	11	eleven
2	two	12	twelve
3	three	13	thirteen
4	four	14	fourteen
5	five	15	fifteen
6	six	16	sixteen
7	seven	17	seventeen
8	eight	18	eighteen
9	nine	19	nineteen

Example 20s to 100

20	twenty	90	ninety
21	twenty-one	91	ninety-one
22	twenty-two	92	ninety-two
23	twenty-three	93	ninety-three
24	twenty-four	94	ninety-four
25	twenty-five	95	ninety-five
26	twenty-six	96	ninety-six
27	twenty-seven	97	ninety-seven
28	twenty-eight	98	ninety-eight
29	twenty-nine	99	ninety-nine
		100	one hundred ok

Summary

The programming took more effort than I anticipated.

Win32Forth has no string concatenate. That was the highest effort task.

There are many irregularities in low numbers. Our children learn by rote.

How do non-English speakers learn?

For voice response, the hyphen '-' would be omitted.

